Plot for Example 5: x' = 2x+y^2, y' = x + y + xy.

Most of the code below is the same as used in Example 1 under Section 9.3.

```
> with(DEtools):with(plots):with(plottools): # Load various
  packages
> # First, I mark the three critical points with small green disks.
> R:=disk([0,0],0.05,color=green):
> S1:=disk([-1/2,1],0.05,color=green):
> S2:=disk([-2,-2],0.05,color=green):
> # Next I create six black line segments for the eigenspaces of
  the linearizations at each critical point.
> RES1:=line([-1/2, -1/2], [1/2, 1/2], color = black):
> RES2:=line([0, -1/2], [0, 1/2], color = black,thickness=2):
> S2ES1:=line([-2-1/2, -2-1/2], [-2+1/2,-2+1/2], color = black):
> S2ES2:=line([-2-2/sqrt(17), -2+1/2/sqrt(17)], [-2+2/sqrt(17),-2
  -1/2/sqrt(17)], color = black):
> S1ES1:=line([-1/2+1/2,1+(sqrt(73)-3)/16],[-1/2-1/2,1-(sqrt(73)-3)
  /16],color=black):
> S1ES2:=line([-1/2+1/4,1+(-sqrt(73)-3)/32],[-1/2-1/4,1-(-sqrt(73)
  -3)/32],color=black):
> # Then I create the solutions for several initial conditions.
> solutioncurves:=phaseportrait([D(x)(t) = 2*x(t) + (y(t))^2, D(y)
  (t)= x(t) + y(t) + x(t)*y(t)],[x(t),y(t)],t=0..5, [[x(0)=-2.2,y
  (0)=3],[x(0)=-2.1,y(0)=3],[x(0)=-2.5,y(0)=3],[x(0)=-2,y(0)=3],[x
  (0)=-1.0,y(0)=-1.1],[x(0)=-2.7,y(0)=-2.8],[x(0)=-2.9,y(0)=-2.8],
  [x(0)=0.1,y(0)=0.1],[x(0)=0.0,y(0)=0.1],[x(0)=-0.1,y(0)=-0.1],[x
  (0)=-0.1,y(0)=0.0],[x(0)=0.0,y(0)=-0.1],[x(0)=-0.1,y(0)=0.5],[x
  (0)=-0.2,y(0)=0.4]],x=-3..3,y=-3..3,linecolor=red,arrows=none,
  stepsize=0.02):
> # Then I create the vector field.
> vectorfield:=fieldplot([2*x+y^2,x+y+x*y],x=-3..3,y=-3..3,arrows=
  slim,anchor=tail,fieldstrength=maximal(2),grid=[20,20]):
> C1:=circle([0,0],1,color=blue):
> C2:=circle([-1/2,1],1,color=blue):
> C3:=circle([-2,-2],1,color=blue):
> C4:=circle([0,0],3,color=blue):
> C5:=circle([0,0],2,color=blue):
> # Finally, this is all displayed.
```

```
> display(solutioncurves,vectorfield,R,S1,S2,RES1,RES2,S1ES1,S1ES2,
  S2ES1,S2ES2,C1,C2,C3,C4,C5);
```




```
>
```